


# Making the most out of Multilabel Formats



Tasha Chapman

Oregon Department of  
Consumer and Business Services


**SAS is a registered trademark of SAS Institute in the USA and other countries.  
® indicates USA registration.**

# What we want...

Age	Year		Two year total	
	2004	2005	Total	%
Underage (under 14)	-	-	-	0.0
Minors (ages 14-17)	147	124	271	0.6
Adult	22,089	21,700	43,789	99.4
Ages 18-25	3,104	3,091	6,195	14.1
Ages 26-45	10,778	10,666	21,444	48.7
Ages 46-64	7,842	7,567	15,409	35.0
Ages 65+	365	376	741	1.7
Total	22,236	21,824	44,060	100.0

Created using Proc Tabulate and ODS PDF (style=SASweb)

# What we want...





Age	Year		Two year total	
	2004	2005	Total	%
Underage (under 14)	-	-	-	0.0
Minors (ages 14-17)	147	124	271	0.6
Adult	22,089	21,700	43,789	99.4
Ages 18-25	3,104	3,091	6,195	14.1
Ages 26-45	10,778	10,666	21,444	48.7
Ages 46-64	7,842	7,567	15,409	35.0
Ages 65+	365	376	741	1.7
Total	22,236	21,824	44,060	100.0

Created using Proc Tabulate and ODS PDF (style=SASweb)

# Multilabel option in Proc Format

```
proc format;
  value agefmt (multilabel)
    low-13 = 'Underage (under 14) '
    14-17 = 'Minors (ages 14-17) '
    18-high = 'Adult '
    18-25 = '      Ages 18-25 '
    26-45 = '      Ages 26-45 '
    46-64 = '      Ages 46-64 '
    65-high = '      Ages 65+' ;
```



# MLF option in Proc Tabulate

```
ods pdf style=sasweb file='Age.pdf';  
proc tabulate format=comma6. ;  
format age agefmt. ;  
class age /mlf; ←  
class year ;  
table age = ' ' all='Total',  
       year = 'Year' all='Two year total'  
       *(n='Total' pctn<age all>*F=6.1)  
       /misstext='-' box='Age';  
keylabel pctn='% ' n=' ' ;  
run ;  
ods pdf close ;
```




# Results so far...

- Multilabel format honored
- Rows out of order\*
- Rows with missing data not displayed
- Leading spaces deleted

Age	Year		Two year total	
	2004	2005	Total	%
Ages 18-25	3,104	3,091	6,195	14.1
Ages 26-45	10,778	10,666	21,444	48.7
Ages 46-64	7,842	7,567	15,409	35.0
Ages 65+	365	376	741	1.7
Adult	22,089	21,700	43,789	99.4
Minors (ages 14-17)	147	124	271	0.6
Total	22,236	21,824	44,060	100.0

\*“Adult” came after “Ages” because of the leading spaces in our original format.

# Notsorted option in Proc Format



```
proc format;
  value agefmt (multilabel, notsorted)
    low-13 = 'Underage (under 14)'
    14-17 = 'Minors (ages 14-17)'
    18-high = 'Adult'
    18-25 = '    Ages 18-25'
    26-45 = '    Ages 26-45'
    46-64 = '    Ages 46-64'
    65-high = '    Ages 65+' ;
```

A red arrow points from the top right towards the `notsorted` option in the `value` statement, which is also circled with a red dashed line.

# Preloadfmt option in Proc Tabulate

```
ods pdf style=sasweb file='Age.pdf';  
proc tabulate format=comma6.;  
format age agefmt.;  
class age /mlf preloadfmt order=data;  
class year /order=internal;  
table age = ' ' all='Total',  
       year = 'Year' all='Two year total'  
       *(n='Total' pctn<age all>*F=6.1)  
       /misstext='-' box='Age';  
keylabel pctn='% ' n=' ';  
run;  
ods pdf close;
```



# Order= options in Proc Tabulate

```
ods pdf style=sasweb file='Age.pdf';  
proc tabulate format=comma6.;  
format age agefmt.;  
class age /mlf_preloadfmt order=data;  
class year /order=internal;  
table age = 'Total',  
       year = 'Year' all='Two year total'  
       *(n='Total' pctn<age all>*F=6.1)  
       /misstext='-' box='Age';  
keylabel pctn='% ' n='';  
run;  
ods pdf close;
```

# Results so far...

- Multilabel format honored
- Rows in order
- Rows with missing data not displayed
- Leading spaces deleted

Age	Year		Two year total	
	2004	2005	Total	%
Minors (ages 14-17)	147	124	271	0.6
Adult	22,089	21,700	43,789	99.4
Ages 18-25	3,104	3,091	6,195	14.1
Ages 26-45	10,778	10,666	21,444	48.7
Ages 46-64	7,842	7,567	15,409	35.0
Ages 65+	365	376	741	1.7
Total	22,236	21,824	44,060	100.0

# Rows with missing data

```
proc format;  
  value agefmt (multilabel notsorted)  
  low-13 = 'Underage (under 14)'  
  14-17 = 'Minors (ages 14-17)'  
  18-high = 'Adult'  
  18-25 = '    Ages 18-25 '  
  26-45 = '    Ages 26-45 '  
  46-64 = '    Ages 46-64 '  
  65-high = '    Ages 65+' ;
```

# Printmiss option in Proc Tabulate

```
ods pdf style=sasweb file='Age.pdf';  
proc tabulate format=comma5.;  
format age agefmt.;  
class age /mlf preloadfmt order=data;  
class year /order=internal;  
table age = ' ' all='Total',  
       year = 'Year' all='Two year total'  
       *(n='Total' pctn<age all>*F=6.1)  
       /misstext='-' box='Age' printmiss;  
keylabel pctn='% ' n=' ';  
run;  
ods pdf close;
```


# Results so far...

- Multilabel format honored
- Rows in order
- Rows with missing data displayed
- Leading spaces deleted

Age	Year		Two year total	
	2004	2005	Total	%
Underage (under 14)	-	-	-	0.0
Minors (ages 14-17)	147	124	271	0.6
Adult	22,089	21,700	43,789	99.4
Ages 18-25	3,104	3,091	6,195	14.1
Ages 26-45	10,778	10,666	21,444	48.7
Ages 46-64	7,842	7,567	15,409	35.0
Ages 65+	365	376	741	1.7
Total	22,236	21,824	44,060	100.0

# Leading spaces in Proc Format

```
proc format;
  value agefmt (multilabel notsorted)
    low-13 = 'Underage (under 14)'
    14-17 = 'Minors (ages 14-17)'
    18-high = 'Adult'
    18-25 = '    Ages 18-25'
    26-45 = '    Ages 26-45'
    46-64 = '    Ages 46-64'
    65-high = '    Ages 65+' ;
```



# ASIS style option in Proc Tabulate

```
ods pdf style=sasweb file='Age.pdf';

---

proc tabulate format=comma6. ;  
format age agefmt. ;  
class age /mlf preloadfmt order=data ;  
class year /order=internal ;  
classlev age /style={asis=yes} ;  
table age = _ _ _ all='Total' ,  
          year = 'Year' all='Two year total'  
          * (n='Total' pctn<age all>*F=6.1)  
          /misstext='-' box='Age' printmiss ;  
keylabel pctn='% ' n=' ' ;  
run ;

---

ods pdf close ;
```

# Final Results


- Multilabel format honored
- Rows in order
- Rows with missing data displayed
- Leading spaces maintained

Age	Year		Two year total	
	2004	2005	Total	%
Underage (under 14)	-	-	-	0.0
Minors (ages 14-17)	147	124	271	0.6
Adult	22,089	21,700	43,789	99.4
Ages 18-25	3,104	3,091	6,195	14.1
Ages 26-45	10,778	10,666	21,444	48.7
Ages 46-64	7,842	7,567	15,409	35.0
Ages 65+	365	376	741	1.7
Total	22,236	21,824	44,060	100.0



# Tasha Chapman

Oregon Department of  
Consumer and Business Services



Contact info:

Tasha.L.Chapman@state.or.us

**Paper and presentation available at:**  
**[www.sasCommunity.org](http://www.sasCommunity.org)**